HRP 259 SAS EG LAB FIVE

Lab Five: Data exploration exercise: linear regression

Lab Objectives

After today’s lab you should be able to:

1. Explore a dataset to identify outliers and missing data.

2. Plot data distributions. 

3. Check for normality.

4. Obtain Pearson’s correlation coefficients between multiple covariates (must be continuous or binary).

5. Build a linear regression model.

6. Dummy code categorical predictors.

7. Use a class statement to make SAS dummy code for you.

8. Look for confounding in the context of linear regression.

9. Walk through a real data analysis exercise.

SAS PROCs                                     

SAS EG equivalent

PROC UNIVARIATE


Describe(Distribution Analysis…
PROC CORR                                                Analyze(Multivariate(Correlations
PROC REG




Analyze(Regression(Linear regression
PROC GPLOT



Graph(Scatter Plot
PROC GLM

                       

Analyze(ANOVA(Linear models

LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Double-click on the SAS icon to open SAS.

2. Goto the class website: www.stanford.edu/~kcobb/courses/hrp259
3. --> Download Lab 5 DATA, runners, which is already in SAS format. Place the dataset on your desktop.

The dataset contains data on 99 runners. Variables include

Outcome:
pchange= percent change in spine bone density since baseline 

Predictors:

weightchg=percent change in weight since baseline
coffee = coffee drinking (cups/day)

stddrink = number of standard drinks of alcohol per day

dairy= dairy intake (servings of dairy/day)

kcal= calorie intake (kcal/day)

soda= soda intake (ounces/day)

ed=history of an eating disorder (yes/no)

veggie=vegetarian (yes/no)

weeklymiles = miles run per week

lifting = weight lifting (minutes/week)

The aim of the study is to test whether coffee drinking affects changes in bone density, controlling for potential confounders.

4. Name a lab5 library using point and click.

To create a permanent library, click on Tools(Assign Project Library…

Type the name of the library, lab5 in the name box. SAS is caps insensitive, so it does not matter whether caps or lower case letters appear. Then click Next.

Browse to find your desktop. We are going to use the desktop as the physical folder where we will store our SAS projects and datasets. Then click Next.

For the next screen, just click Next…

Then click Finish.

5. Use point-and-click to examine the distributions of several variables.

Describe(Distribution Analysis

In the Data screen, drag several variables (including the outcome variable) to “Analysis variables” (
[image: image7.jpg]Al Distribution Analysis1 for Local:LAB5.RUNNERS

[Data

Data
gl

|

oma Detasoce:  LocalLABS AUNNERS

Lopome, [T

Erporeril

W

et Vaiibles o assign T

bk Nene 16 A vaales
o Do G

ovesnce | | @ weitichg @ ooles

Inset @ fesl © sdink

Tablr @ diy of
Tiles @ soda [ Group analysis by

Frequency count [Limit: 1)
Properies @ colles
Relstive wegh (Lnk 1
@ sk E] o (L 1]

{8 Classification variables (Limit 2|
@ed

@ veagee

@ weekimies

@ liting
b _0BSTAT_





In the Distributions screen, check the box beside Normal to get tests for normality:

In the Plots screen ask for histograms and Q-Q plots (for normality)

6. Use the results to answer questions such as:

a. How many subjects are in the study?

b. What is the range of outcome values in the study?

c. Is the outcome variable normally distributed?

d. Can you find any outliers or data entry errors in the data?

7. To get correlations with point-and-click, Analyze(Multivariate(Correlations

Drag all the variables to “Analysis Variables”:

Note: we are requesting Pearson correlations, which is the default. If you wanted to ask for other types of correlation coefficients, you can find these under “Options.”

Go to the Results screen. 1. Under Plots(Create a scatter plot for each correlation pair; 2. check off the “Show correlations in decreasing order of magnitude” box, followed by “Show n correlations per row variable”, and “3” (to get the best 3 per variable).

Then Click Run…
	Pearson Correlation Coefficients 
Prob > |r| under H0: Rho=0 
Number of Observations

	pchange
	pchange

1.00000

 

99


	weightchg

0.30370

0.0022

99


	coffee

0.27769

0.0056

98



	weightchg
	weightchg

1.00000

 

99


	pchange

0.30370

0.0022

99


	weeklymiles

-0.14440

0.1604

96



	kcal
	kcal

1.00000

 

99


	dairy

0.39929

<.0001

99


	veggie

0.23924

0.0177

98



	dairy
	dairy

1.00000

 

99


	kcal

0.39929

<.0001

99


	soda

-0.11483

0.2602

98



	soda
	soda

1.00000

 

98


	weeklymiles

0.18521

0.0724

95


	kcal

-0.15130

0.1370

98



	coffee
	coffee

1.00000

 

98


	ed

0.31143

0.0018

98


	pchange

0.27769

0.0056

98



	stddrink
	stddrink

1.00000

 

98


	weeklymiles

-0.15803

0.1261

95


	dairy

-0.10739

0.2926

98



	ed
	ed

1.00000

 

99


	coffee

0.31143

0.0018

98


	pchange

0.18427

0.0679

99



	veggie
	veggie

1.00000

 

98


	kcal

0.23924

0.0177

98


	pchange

-0.16610

0.1021

98



	weeklymiles
	weeklymiles

1.00000

 

96


	soda

0.18521

0.0724

95


	stddrink

-0.15803

0.1261

95



	lifting
	lifting

1.00000

 

93


	soda

0.14424

0.1701

92


	coffee

-0.11156

0.2897

92




Percent change in bone density is most strongly correlated with weight change. It is also correlated positively with coffee. There is a slight positive correlation with a history of eating disorders, which doesn’t quite reach statistical significance; and a slight negative correlation with vegetable intake that doesn’t reach statistical significance. 

8. FYI, the corresponding code would be… 


proc corr data=lab5.runners best=3;


var weightchg kcal dairy soda coffee


stddrink ed veggie  weeklymiles lifting; 

run;

9. Run a linear regression using point and click with stepwise selection (automatic selection procedure): Analyze(Regression(Linear regression

Drag pchange to the Dependent variable. Drag all the other variables to Explanatory variables.

In the Model screen, choose Stepwise Selection:

Then use .10 as your model entry and exit criteria. Note that, in this screen, you can also choose variables that you would like to force to stay in the model. 

Be wary of the results from automatic selection procedures!! They should only be used for exploratory purposes (e.g., here we just want to quickly determine the strongest predictors of pchange on a p-value basis). Stepwise selection should not be used for testing particular hypotheses, such as the relationship between coffee and bone density changes. 
At the end of your output, find:

	Summary of Stepwise Selection

	Step
	Variable
Entered
	Variable
Removed
	Number
Vars In
	Partial
R-Square
	Model
R-Square
	C(p)
	F Value
	Pr > F

	1
	coffee
	 
	1
	0.0707
	0.0707
	5.4533
	6.62
	0.0118

	2
	weightchg
	 
	2
	0.0663
	0.1370
	0.9998
	6.61
	0.0119





10. FYI, to do this same exploratory linear regression using code:



proc reg data=lab5.runners;

model pchange = weightchg kcal dairy soda coffee


stddrink ed veggie  weeklymiles lifting/selection=stepwise sle=.10;

run;
11. Now visually inspect the scatter plot between coffee and pchange (the main relationship of interest):

Graph(Scatter Plot

Ask for a 2-D plot:

In the Data screen, choose pchange as the vertical axis variable and coffee as the horizontal axis variable: 

In the Interpolations screen, pick Regression and then Linear  to get the linear regression line superimposed on the plot. 

12. Use Modify Task to ask for a smooth line rather than a linear regression line…

In the Interpolations screen, pick Smooth and then 60; the default is to sort the data first, so you do not have to worry about this (leave it checked). 

[image: image25.png]coffee




This plot reveals some interesting things—The effect of coffee appears to begin after 1 cup and to level off after 3 cups. This suggests another way to model coffee drinking—you could compare 0-1 cup drinkers to 2-3 cup drinkers to 3+ drinkers.

CORRESPONDING CODE, FYI:

proc gplot data=lab5.runners;

plot pchange*coffee;

symbol v=dot c=red i=rl;

run;

proc gplot data=lab5.runners;

plot pchange*coffee;

symbol v=dot c=red i=sm60s;

run;
13. Now we’ll do some model building. We’re going to use code because it’s more efficient here…

Start a New Program. Name it “coffee regressions.” Test the hypothesis that coffee drinking is related to bone density change, starting with a univariate regression:

proc reg data=lab5.runners;

model pchange = coffee;

run;
	Parameter Estimates

	Variable
	DF
	Parameter
Estimate
	Standard
Error
	t Value
	Pr > |t|

	Intercept
	1
	0.24478
	0.33066
	0.74
	0.4609

	coffee
	1
	0.49333
	0.17419
	2.83
	0.0056


Univariate model:

Percent change in spine bone density (%) = 0.25 + 0.50 (cups coffee/day)

Interpretation: on average, women increased just 0.25% in spine bone density during the study; each cup of daily coffee was associated with an additional 0.5% increase. 

In univariate regression, coffee is a significant positive predictor of bone density changes.

14. Now write a program to test for confounding by the potential confounders. For example, coffee drinking may be associated with increased dairy intake (e.g., lattes) and increased dairy may explain the link with bone density. 

To test for confounding, include each covariate in the model one at a time with the main predictor and see how inclusion of the covariate affects the relationship between coffee and the outcome (i.e., affects the beta coefficient). To speed up the process of doing this, use a MACRO:


%macro speed(confounder);

proc reg data=lab5.runners;

model pchange = coffee &confounder.;

run;

%mend;

%speed(kcal);

%speed(stddrink);

%speed(dairy);

%speed(weightchg);

%speed(soda);

%speed(ed);

%speed(veggie);

%speed(weeklymiles);

%speed(lifting);

Scroll through the output to find that there is little evidence of confounding. The biggest confounder is a history of eating disorders (ed). Including this variable (ed) changes the point estimate of coffee drinking from a beta of 0.49 to a beta of 0.43. As a rule of thumb, changes greater than 10% are considered confounders; we just barely make this cutoff here.

Why is this a confounder? Women with a history of eating disorders tend to drink more coffee and they have a greater increase in bone density in this sample (probably because they start with greatly diminished bone density and are catching up on their bone mineralization).
15. Now, assemble your final model: include coffee, ed (because of the slight evidence of confounding), and weight change (since weight change is itself a strong predictor).

proc reg data=lab5.runners;

model pchange=coffee ed weightchg;

output out=diagnostics rstudent=residual;

run;
	Parameter Estimates

	Variable
	DF
	Parameter
Estimate
	Standard
Error
	t Value
	Pr > |t|

	Intercept
	1
	0.06986
	0.35639
	0.20
	0.8450

	coffee
	1
	0.48544
	0.17459
	2.78
	0.0066

	ed
	1
	0.50312
	0.59813
	0.84
	0.4024

	weightchg
	1
	0.14374
	0.04247
	3.38
	0.0010


Final model: 

Change in spine bone density (%) = 0.06 + 0.485*(cups coffee/day) +0.50*(1 if history of an eating disorder) + 0.14*(% change in weight)

For example, study subject 1 has pchange of 2.5%, weight change of -.29%, coffee of 2.0 cups/day, and ed=1. So, her predicted change in bone density is:

Predicted change in spine bone density (%) = 0.06 + 0.485*(2.0) +0.50*(1) + 0.14*(-.29) = 1.50
Residual = observed-predicted = 2.5-1.5= 1.0

16. Plot the residuals against each predictor to look for patterns; we’ll use code here to practice using the code for PROC GPLOT…
axis1 label=(angle=90);

proc gplot data=diagnostics;

plot residual*coffee /vaxis=axis1;

symbol v=dot c=red i=sm60s;

run;

17. As I mentioned before, you may also want to try coffee as a categorical predictor, because it looks like the pattern is not strictly linear. To model coffee as categorical, create dummy codes for the different categories:

data runners;

set lab5.runners;

if 1<coffee<=3 then medcoff=1; else medcoff=0;

if coffee>3 then highcoff=1; else highcoff=0;

run;

proc freq data=runners;

tables medcoff highcoff;

run;


	medcoff
	Frequency
	Percent
	Cumulative
Frequency
	Cumulative
Percent

	0
	87
	87.88
	87
	87.88

	1
	12
	12.12
	99
	100.00


	highcoff
	Frequency
	Percent
	Cumulative
Frequency
	Cumulative
Percent

	0
	87
	87.88
	87
	87.88

	1
	12
	12.12
	99
	100.00



proc reg data=runners;

model pchange=medcoff highcoff ed weightchg;

run;
	Parameter Estimates

	Variable
	DF
	Parameter
Estimate
	Standard
Error
	t Value
	Pr > |t|

	Intercept
	1
	0.13361
	0.35257
	0.38
	0.7056

	medcoff
	1
	0.11088
	0.85671
	0.13
	0.8973

	highcoff
	1
	2.58672
	0.86836
	2.98
	0.0037

	ed
	1
	0.62527
	0.59264
	1.06
	0.2941

	weightchg
	1
	0.13668
	0.04207
	3.25
	0.0016


Interpretation: the lowest group increases 0.13% per year (intercept); the medium coffee group increases 0.13+0.11=0.24% per year, but this is not significantly different than the lowest group (p=.89); the highest group increases 0.13+2.59=2.72% per year, which is significantly different than the lowest group (p=.0037). 

18. The other way to do categorical predictors is to let SAS do the dummy coding for you. In PROC GLM, if you put a categorical variable on the class line, SAS will automatically dummy code it:

data runners;

set lab5.runners;

if coffee<=1 then coffeec="clow";

else if coffee<=3 then coffeec="bmed";

else if coffee>3 then coffeec="ahigh";

run;

proc glm data=runners;

class coffeec ;

model pchange=coffeec ed weightchg/solution;

lsmeans coffeec /pdiff adjust=tukey;

run;

	Parameter
	Estimate
	 
	Standard Error
	t Value
	Pr > |t|

	Intercept
	0.133614310
	B
	0.35256857
	0.38
	0.7056

	coffeec ahig
	2.586722111
	B
	0.86835631
	2.98
	0.0037

	coffeec bmed
	0.110880168
	B
	0.85671217
	0.13
	0.8973

	coffeec clow
	0.000000000
	B
	.
	.
	.

	ed
	0.625270939
	 
	0.59264305
	1.06
	0.2941

	weightchg
	0.136681924
	 
	0.04206991
	3.25
	0.0016


NOTE: The X'X matrix has been found to be singular, and a generalized inverse was used to solve

      the normal equations.  Terms whose estimates are followed by the letter 'B' are not

      uniquely estimable. 


	coffeec
	pchange LSMEAN
	LSMEAN Number

	ahig
	2.94068176
	1

	bmed
	0.46483982
	2

	clow
	0.35395965
	3


	Least Squares Means for effect coffeec
Pr > |t| for H0: LSMean(i)=LSMean(j)
Dependent Variable: pchange

	i/j
	1
	2
	3

	1
	 
	0.0683
	0.0102

	2
	0.0683
	 
	0.9908

	3
	0.0102
	0.9908
	 








































































































Same model as before.





The high coffee group is significantly different from the low group and borderline (p=.06) from the medium, after adjusting for multiple comparisons 





Least squares means give the mean predicted value for % change in bone density in each group, at the average values of ed and weight change (based on the regression model). 





You may ignore this warning. 





To get p-values for all pairwise comparisons.





In PROC GLM, identify categorical variables on the CLASS line. SAS dummy codes for you.





Create a categorical variable for coffee (three levels). I want the lowest group to be the reference, so it must be alphabetically last.





Leave the lowest group as your reference group.





12 in the high and med groups, a little small…





The lowest group is left as the reference group.





Make sure all your groups are big enough to make reasonable estimates.





That red line can mislead your eye, so use i=sm60s to ask for a curved line, which better captures more subtle patterns.





sm: asks for smoothed curve


60: you can put any number from 01 to 99 to tell SAS how much curvature you want. If you ask for “01” you get a very zig-zagged line; if you ask for “99” you get a straight line. 60 asks for something in between.


s=sort first (this is important! or you will end up with a big mess)








We will do a lot more on graphing later; but, for now, the symbol statement changes the plotting symbol. Here I’m asking SAS to plot with a red dot, and to superimpose a linear regression line (i=rl says “interpolate” or connect the points with a linear regression line).





Rather than writing a PROC REG out for each potential confounder (or doing multiple point and clicks!), simplify your code by writing a single macro and then applying it to each potential confounder.





Ask for residuals so you can check the fit of your final model.





rstudent asks for “studendized” residuals (gives a Z value rather than absolute value).








Restricts correlation matrix to the top 3 correlations for each predictor.





Setting the axis allows us to turn the vertical axis label by 90 degrees.





Recall:” &variable. is a macro variable.





Calorie intake is strongly correlated with the total number of daily servings of dairy foods/beverages. (r=.399, p<.0001)








Percent change in bone density is more correlated with weight change and coffee drinking.





Asks SAS to use stepwise selection to find the predictors that most strongly predict change in bone density (i.e., smallest p-values). 


**NOTE: Automatic selection are purely exploratory; they should NOT be used for producing the final model, especially when our goal is to test a particular hypothesis.





These two variables held up as the strongest predictors of bone density changes in this sample. They are listed in order of their predictive strength.





Partial R-square tells you how much of the variability in the outcome is explained by the predictor variable, after controlling for previously listed predictors. So, alone, coffee drinking explains 7.1% of the variance in bone density changes. After taking into coffee drinking, weight change explains an additional 6.6% of the variance. 





p-value for the beta-coefficient for each predictor. 


Null hypothesis: beta=0





PROC REG is the procedure for basic linear regression.  








sle=.10 tells SAS to only enter variables in the model if they have a p-value of <=.10; and remove them from the model if their p-value increases to >.10 when combined with the other variables. 

